
  各種方程式を解くためのツールとして、表１に示す５種類のソルバーを提供します。各

ソルバーのプログラムは、方程式データを設定して計算できる汎用仕様になっています。

また、ソルバー計算部分はサブルーチンにしてありますので、他のプログラム開発に利用す

ることも可能です。なお、データファイルの文字コードはUTF-8に統一しています。

                         表１    添付のソルバー一覧

ソルバー名称（VBﾌﾟﾛｼﾞｪｸﾄ名） ソルバーの概要

[1] フルバンド１次方程式 

   （BMSolverF）

多元１次の連立方程式において、係数行列がバンドマ

トリックスとなる場合にフルバンドで解くもの。

[2] ハーフバンド１次方程式

   （BMSolverH）

多元１次の連立方程式において、係数行列が対称バン

ドマトリックスとなる場合にハーフバンドで解くも

の。

[3] 多元１次方程式

   （LineEqua）

多元１次の連立方程式について、ガウス消去法で解く

もの。

[4] １元高次方程式

   （PowerEqua）

１元高次方程式について、Regular Falsi法を用いて実

数解を算出するもの。

[5] 常微分方程式

   （OrdDiffEqua）

常微分方程式について、Runge Kutta Gill法を用いて

解を算出するもの。

第1章 フルバンド１次方程式

1.1 バンドマトリックス計算

  差分法や有限要素法は、１つの部分要素の方程式を作成し、それを全体にまとめて計算す

る方法であり、そのまとめた全体行列を解いて答を算出します。その際、全体行列は、図1-1

に示すように、対角の一定幅の部分だけに数値があり、それ以外の部分は０です。なお、

行列方程式は次式で表わします。

          Ａ ｘ ＝ ｂ                                      (1-1)

          ただし、Ａ：全体行列

                  ｘ：未知数の列ベクトル

                  ｂ：右辺ベクトル

                              図1-1    行列［Ａ］の形

  図1-1のような行列Ａを持つ方程式(1-1)を解くには、コンピューターのメモリー容量と計

算速度の点から、数値が０の右上や左下の部分を除外して、行列の対角の一定幅の部分だけ

を取り出して計算します。これがバンドマトリックス計算です。ここで、バンドとは、対角

の一定幅の部分が帯状になっているので、そのように呼んでいます。そして、バンドの一定

幅をバンド幅と呼びますが、図1-1のように、バンド全体の幅がフルバンド幅WFです。流れ解

析等では上記の行列［Ａ］が非対称なため、フルバンドを用いて方程式を解きます。

  他方、通常の構造解析や熱伝導解析では、行列［Ａ］が、対角要素に対して、右上側と左

下側が対称な行列になるので、行列の右上側のみを用いて計算できます。その場合は、対角

要素とその右側の一定幅の部分だけを用いればよいので、メモリー容量と計算速度の点で、

一層有利になります。その際、対角要素とその右側の一定幅の部分をハーフバンドと呼び、

方程式ソルバーK  2

＊ ＊ ＊ ＊ ＊

＊ ＊ ＊ ＊ ＊

＊ ＊ ＊ ＊ ＊

＊ ＊ ＊ ＊ ＊

＊ ＊ ＊ ＊ ＊

＊ ＊ ＊ ＊ ＊

＊ ＊ ＊ ＊ ＊

＊ ＊ ＊ ＊ ＊

＊

＊ ＊

＊ ＊

＊ ＊ ＊ ＊

＊

＊＊ ＊ ＊

＊＊ ＊ ＊

＊＊ ＊ ＊

＊＊ ＊ ＊

＊＊ ＊ ＊

＊＊ ＊ ＊

＊＊ ＊ ＊

＊＊ ＊ ＊

＊ ＊ ＊

＊ ＊

＊

０

０

フルバンド幅 W F

ハーフバンド幅 W H



その幅が図1-1に示すハーフバンド幅WHです。

  差分法・有限要素法では、バンドマトリックスを用いることが常識ですが、初心者の方の

ため、バンドマトリックスがどれくらい有利なのか参考として以下に例を示します。たとえ

ば小さなモデルで、未知数の数が200、フルバンド幅が53とすると、

        ［全体行列の配列要素数］200行×200列＝40000

        ［フルバンドマトリックスの配列要素数］200行×53列＝10600

        ［ハーフバンドマトリックスの配列要素数］200行×27列＝5400

  このように、バンドマトリックスを用いれば、必要な配列要素数を大幅に低減でき、上例

のハーフバンドマトリックスでは全体行列の13.5％にまでなっています。そして、配列要素

数を低減すれば、それに比例して、使用メモリー容量が少なくてすみ、計算時間も短縮でき

ます。

  図1-2にフルバンドマトリックスの計算方法を示します。

                   図1-2    フルバンドマトリックスの計算方法

  図1-2に示すように、先ず前進計算で、バンドの左下側の要素をすべて０にするように、第

１行目の対角要素から順に計算していきます。そして、最終行の１つ前の対角要素まで計算

したら、左下側の要素がすべて０になり、最終行の未知数ｘnがその右辺の値から算出できま

す。そうすれば、最終行の１つ前の行からｘn-1を算出でき、順次、１行ずつ上側へ後退計算

させ、第１行まですべてのｘを求めることができます。

  1)フルバンドのマトリックス計算プログラム

      図1-3にフルバンドマトリックスを解くプログラムを示します。同図で、○囲み数字の

    部分について以下に補足説明します。

      ①サブルーチンの引数は、エラー有無kh、ハーフバンド幅mh、変数の数nx、係数行列

        fmat(,)、右辺兼解cvev()です。

          ◇サブルーチンを呼び出す場合は、次のように記述します。

                Call SolverFull(kh, mh, nx, fmat, cvec)

    　②マトリックスの不在列チェックを行なっています。すなわち、最後の方の行におい

        ては、全体行列の最終列より右側までバンドマトリックスに含まれるので、計算

        する列数mxは、その不在列数分だけ減らします。

      ③pivotvalueが0であると、以降の割算の際にゼロ割算のエラーとなります。通常は0

        になりませんが、もし0が生じる場合はエラー処理が必要です。

      ④ピボットの行列要素（j=1）は、pivotvalueで割ると1になりますが、後の計算で

        不要なためピボット要素の置換は省略しています。

      ⑤ピボット行より下行のピボット列を0とする演算を行なっています。ピボット列

        を0にするのは、方程式演算により行なうので、ピボット列以外の各要素も演算

        変更されます。

      ⑥(nx-1)行から1行まで遡ります。

      ⑦前述の①と同様にマトリックスの不在列チェックを行なっています。すなわち、

        最後の方の行においては、計算する列数mxは、その不在列数分だけ減らします。

      ⑧現在列よりも、右側の変数は先に解が求まっているので、それらの変数項を計算   

        できます。

      ⑨右辺cvec(k)からsを引いた解をcvec(k)に入れています。

              図1-3（1/2）    フルバンドマトリックス計算プログラム

Sub SolverFull(ByRef kh As Integer, ByVal mh As Integer, ByVal nx As Integer, _

ByRef fmat(,) As Double, ByRef cvec() As Double)                          ①

    'ﾌﾙﾊﾞﾝﾄﾞ･ﾏﾄﾘｯｸｽのｿﾙﾊﾞｰ

    'kh･･･ｴﾗｰ有無（0:なし、-1:有り）

    'mh･･･ハーフバンド幅

    'nx･･･変数の数

    'fmat(,)･･･フルバンドの係数行列

    'cvec()･･･右辺および解

    Dim aa$

    Dim i, it, j, jk, jt, k, mx As Integer

    Dim pivotvalue, s, valuex As Double

    kh = 0

    '前進方向の計算

    For k = 1 To nx - 1

        mx = mh

        If k + mh - 1 > nx Then   'ﾏﾄﾘｯｸｽの不在列ﾁｪｯｸ                     ②

            mx = nx - k + 1

        End If

        pivotvalue = fmat(k, mh)

        If pivotvalue = 0 Then    'ｴﾗｰ処理                                ③

            kh = -1

            aa$ = "pivotvalue=0 AT k=" + Str$(k)

            aa$ = aa$ + Chr(13) + Chr(10) + "計算を終了させます！"

            MsgBox(aa$, 0, "計算エラー")

            Exit Sub

        Else

            For j = 2 To mx                                               ④

                jk = mh + j - 1

                fmat(k, jk) = fmat(k, jk) / pivotvalue

            Next j

        End If

        cvec(k) = cvec(k) / pivotvalue

        For i = 2 To mx                                                   ⑤

            it = i + k - 1

            jk = mh - i + 1

            valuex = fmat(it, jk)

            If valuex <> 0 Then

                For j = 2 To mx

                    jt = j - i + mh

                    jk = mh + j - 1

                    fmat(it, jt) = fmat(it, jt) - valuex * fmat(k, jk)

                Next j

                cvec(it) = cvec(it) - valuex * cvec(k)

            End If

        Next i

    Next k



              図1-3（2/2）    フルバンドマトリックス計算プログラム

1.2 プログラムの機能・構成

      ［ＶＢプロジェクト名］ＢＭＳｏｌｖｅｒＦ

  本ソフトは、多元１次の連立方程式において、係数行列がバンドマトリックスとなる場合

にフルバンドで解くものです。本ソフトの立ち上げ画面は図1-4のメニュー画面です。同図の

ように、基本メニューは「データ」、「計算」、「終了」の３つです。「データ」メニュー

では入力データを作成し、「計算」メニューでは方程式計算を実行します。そして、ソフト

を終了させる際は、「終了」をクリックします。

  次に、図1-5に本ソフトのプログラム構成を示します。同図のように、全体を「メニュー」

のSolvFMenuモジュール* で制御し、その下に「データ」および「計算」の２つのプログラ

ム群に分かれています。

    注）*モジュール名と各フォーム名は、ファイル名と名前を一致させてあります。

  データファイルは、「入力データファイル」と計算の「結果データファイル」の２種類を

使用しています。ただし、本図には示していませんが、「計算」部分で一時記憶ファイル

（ファイル拡張子「.tmp」）も使用しています。これらデータファイルはいずれもテキスト

形式であり、エディターや表計算ソフトで閲覧・印刷することができます。

    '最終変数の解
    cvec(nx) = cvec(nx) / fmat(nx, mh)

    '後退方向の計算
    For k = nx - 1 To 1 Step -1                                            ⑥
        s = 0
        mx = mh
        If k + mh - 1 > nx Then                                            ⑦
            mx = nx - k + 1
        End If
        For j = 2 To mx                                                    ⑧
            jt = j + k - 1
            jk = mh + j - 1
            s = s + fmat(k, jk) * cvec(jt)
        Next j
        cvec(k) = cvec(k) - s
    Next k

End Sub



                図1-4    「ＢＭＳｏｌｖｅｒＦ」ソフトのメニュー画面

              図1-5    「ＢＭＳｏｌｖｅｒＦ」ソフトのプログラム構成

  表1-1に本ソフトのファイルリストを示します。同表で示すように、［データ］の

SolvFDat1～SolvFDat2は、各種データを画面入力するためのものです。特に、SolvFDat2で

は、行列データをグリッド入力できるようにしています。

               表1-1    「ＢＭＳｏｌｖｅｒＦ」ソフトのファイルリスト

             ［全般］

ファイル名 内  容

SolvFCom.vb 共通変数の宣言とサブルーチンの定義

SolvFMenu.vb メニューの提示･選択

             ［データ］
          

ファイル名 内  容

SolvFDat1.vb データ仕様の入力

SolvFDat2.vb 行列データの入力

AnalDat6.vb データファイルの入出力

             ［計算］
          

ファイル名 内  容

SolvFCalc.vb 使用するデータファイルの設定

SolvFCal3.vb 方程式計算の実行

1.3 例題

  次節以降では、次の例題のデータを使用して説明します。

      Ａ=[
2 3 4 0 0 0 0 0
5 6 7 8 0 0 0 0
9 10 12 13 14 0 0 0
0 15 16 17 18 19 0 0
0 0 20 21 22 23 24 0
0 0 0 25 26 27 28 29
0 0 0 0 30 31 32 33
0 0 0 0 0 34 35 36

]                     (1-2)

      ｂ={
10
20
30
40
50
60
70
80

}                                              (1-3)

1.4 「データ」処理プログラム

1.4.1 サブメニュー画面

    「メニュー」プログラムにおける「データ」のサブメニュー選択画面を図1-6に示します。

  同図に示すように、５つのサブメニューがあります。全く新たにデータを作成する場合は

  「データ入力・修正」を選択しますが、既存ファイルを修正して新たなファイルを作成す

  る場合は先に「ファイル読み込み」を行ないます。

    また、「オプション」→「バックアップファイル作成」にチェックを入れると、データ

  読み込み時にバックアップファイル（.bak）が自動作成されます。チェックを外すと作成

  されなくなります。

                        図1-6    「データ」サブメニュー画面

1.4.2 ファイル読み込み

    サブメニューの「ファイル読み込み」を選択すると、コモンダイアログでファイルを開

  く画面となり、入力データファイルを選定して読み込みます。なお、入力データファイル

  のデフォールトの拡張子は「.dat」です。

1.4.3 データ入力・修正

    サブメニューの「データ入力・修正」を選択すると、図1-7のデータ仕様入力画面となり

  ます。本画面の入力について、以下に補足説明します。

    ①タイトルは、任意の題名を入力します。

    ②変数の数とハーフバンド幅を入力します。

        ◇ハーフバンド幅は対角を含む片側バンド幅です（図1-1参照）。

    ③本画面の「設定」ボタンをクリックすると、次の方程式データ入力画面に移ります。

    ④各データ入力が終了すると本画面に戻るので、そこで「すべて確定」ボタンを

      クリックすると入力データが確定します。

                           図1-7    データ仕様入力画面

1.4.4 方程式データ入力画面

    図1-8に方程式データの入力画面を示します。

    ①係数行列の各要素と右辺の値を入力します。

　　      ◇フルバンドの部分だけを入力します。*印の部分は入力が無効になります。

    ②「コピー」、「貼り付け」ボタンにより、通常のクリップボード利用ができます。

          ◇「取り消し」ボタンを押すと直前の貼り付けを取り消せます。

    ③すべての行列要素データを入力し終ったなら、「設定」ボタンをクリックして図1-7の

      データ仕様入力画面に戻ります。

                          図1-8    方程式データ入力画面

1.4.5 ファイルへの保存

    サブメニューの「ファイルへの保存」を選択すると、コモンダイアログでファイルを

  保存する画面となり、入力データファイルを設定するとデータが書き出されます。新たな

  ファイル名を入力すると新ファイルが作成されます。先に入力データファイルを読み込ん

  である場合に、同ファイルを選定すると上書きとなります。

    図1-9に本例題の入力データファイルを示します。同図のように、１行目のタイトル以

  降、本例題のデータが書き込まれています。

      ◇１行内の複数データは、「,」（コンマ）で区切られます。

      ◇各行でフルバンド幅の係数を書き出しているため、最初の方の行では列0以下の部分

        が0になっており、最後の方の行では列Nを超える部分が0になっています。

                         図1-9    入力データファイル例

1.4.6 最近使用したファイルの選択

    サブメニュー「最近使用したファイル」のポップアップで、ファイルリストから

  選択できます。

  注）新版ソフトでファイルが使用されないとリスト表示されません。

プログラム名,BMSolverF

フルバンド方程式のテストデータ

変数の数, 8

ハーフバンド幅, 3

＜方程式データ＞

行番号、フルバンドの各係数、右辺

 1,   0,   0,   2,   3,   4,    10

 2,   0,   5,   6,   7,   8,    20

 3,   9,   10,   12,   13,   14,    30

 4,   15,   16,   17,   18,   19,    40

 5,   20,   21,   22,   23,   24,    50

 6,   25,   26,   27,   28,   29,    60

 7,   30,   31,   32,   33,   0,    70

 8,   34,   35,   36,   0,   0,    80



1.5 「計算」プログラム

1.5.1 サブメニュー画面

    「メニュー」プログラムにおける「計算」のサブメニュー画面を図1-10に示します。

  サブメニューは「計算の実行」と「計算結果の再表示」の２つがあります。「計算の実

  行」をクリックすると、次の「計算」ファイル設定画面に移ります。また、「計算結果の

  再表示」は、計算の実行後にいつでも結果データを再表示するものです。

                        図1-10    「計算」サブメニュー画面

1.5.2 「計算」ファイル設定

    図1-11に「計算」ファイル設定画面を示します。

    ①「入力ファイル指定」ボタンをクリックすると、コモンダイアログで入力データファ

      イルを選定できます。そして、再び「計算」ファイル設定画面に戻ると、「入力ファ

      イル指定」のテキストボックスには選定済み入力ファイル名が、また「出力ファイル

      指定」の方には入力ファイル名の拡張子を「.rep」に変更したファイル名が表示され

      ます。もし、他のファイル名に変更したい場合は、テキストボックスで修正します。

    ②本画面で「設定」ボタンをクリックすると、計算が開始されます。

                          図1-11    使用ファイル設定画面

1.5.3 計算終了

    図1-12に計算終了画面、図1-13に計算確認ダイアログ画面を示します。図1-12では、か

  かった計算時間も表示されますが、本例では0sとなっており、1s未満で終了しています。

  また、図1-13の計算確認ダイアログ画面では、各変数解も表示されます。本画面で

  「ＯＫ」ボタンをクリックすると、立ち上げ時のメニュー画面に戻ります。

      ◇図1-13の各変数解の表示は、最大10変数までで他は省略されます。

                             図1-12    計算終了画面

                          図1-13    計算確認ダイアログ画面

1.5.4 計算結果

    上記1.5.2で指定した出力ファイルに計算結果が書き出されます。図1-14に例題の計算結

  果ファイルへの出力を示します。

                     図1-14    計算結果ファイルへの出力例

＜＜フルバンドの方程式計算の結果＞＞

フルバンド方程式のテストデータ

［入力データファイル］,C:\VB2022\方程式ｿﾙﾊﾞｰ\BMSolverF\Data\BMSolvFTest.dat

------------------------------------------------------------

＜方程式の解＞

変数番号、   解

 1,   -1.743024

 2,    7.368587

 3,   -2.154929

 4,   -.0514881

 5,   -.1050122

 6,   -1.751812

 7,    5.699241

 8,   -1.664217

------------------------------------------------------------

END    0h 0min 0s



第2章 ハーフバンド１次方程式

2.1 ハーフバンドのマトリックス計算プログラム

  バンドマトリックスが対角要素に関して対称な場合に、ハーフバンド幅で計算することが

できます。図2-1にハーフバンドマトリックスを解くプログラムを示します。基本的にはフル

バンドと同じ計算ですが、バンドマトリックスの配列がハーフバンドの列数しか持たないた

め、計算時に一時的な保存配列ua(mh)を利用しています。フルバンドとの相違点のみ、図2-1

の○囲み数字の部分について以下に補足説明します。                                   

    ①サブルーチンの引数は、エラー有無kh、ハーフバンド幅mh、変数の数nx、係数行列

      fmat(,)、右辺兼解cvev()です。

          ◇サブルーチンを呼び出す場合は、次のように記述します。

                Call SolverHalf(kh, mh, nx, fmat, cvec)

　　②バンドマトリックスのピボット行配列要素については、pivotvalueで割った値をua(j)

      に一時保存し、元の配列要素は変更しません。

    ③バンドマトリックスの配列要素は、ピボット割算されずに残っているので、既知変数

      項の和sをピボット値fmat(k,1)で割っています。

              図2-1（1/2）    ハーフバンドマトリックス計算プログラム

Sub SolverHalf(ByRef kh As Integer, ByVal mh As Integer, ByVal nx As Integer, _

ByRef fmat(,) As Double, ByRef cvec() As Double)                           ①

    'ﾊｰﾌﾊﾞﾝﾄﾞ･ﾏﾄﾘｯｸｽのｿﾙﾊﾞｰ

    'kh･･･ｴﾗｰ有無（0:なし、-1:有り）

    'mh･･･ハーフバンド幅

    'nx･･･変数の数

    'fmat(,)･･･ハーフバンドの係数行列

    'cvec()･･･右辺および解

    Dim aa$

    Dim i, it, j, jt, k, mx As Integer

    Dim pivotvalue, s, valuex As Double

    Dim ua(mh) As Double        '計算途中の保存配列

    kh = 0

    '前進方向の計算

    For k = 1 To nx - 1

        mx = mh

        If k + mh - 1 > nx Then   'ﾏﾄﾘｯｸｽの不在列ﾁｪｯｸ

            mx = nx - k + 1

        End If

        pivotvalue = fmat(k, 1)



             図2-1（2/2）    ハーフバンドマトリックス計算プログラム

        If pivotvalue = 0 Then    'ｴﾗｰ処理

            aa$ = "pivotvalue=0 AT k=" + Str$(k)

            aa$ = aa$ + Chr(13) + Chr(10) + "計算を終了させます！"

            MsgBox(aa$, 0, "計算エラー")

            Exit Sub

        Else

            For j = 2 To mx

                ua(j) = fmat(k, j) / pivotvalue                   ②

            Next j

        End If

        cvec(k) = cvec(k) / pivotvalue

        For i = 2 To mx

            it = i + k - 1

            valuex = fmat(k, i)

            If valuex <> 0 Then

                For j = i To mx

                    jt = j - i + 1

                    fmat(it, jt) = fmat(it, jt) - valuex * ua(j)

                Next j

                cvec(it) = cvec(it) - valuex * cvec(k)

            End If

        Next i

    Next k

    '最終変数の解

    cvec(nx) = cvec(nx) / fmat(nx, 1)

    '後退方向の計算

    For k = nx - 1 To 1 Step -1

        s = 0

        mx = mh

        If k + mh - 1 > nx Then

            mx = nx - k + 1

        End If

        For j = 2 To mx

            jt = j + k - 1

            s = s + fmat(k, j) * cvec(jt)

        Next j

        cvec(k) = cvec(k) - s / fmat(k, 1)                                ③

    Next k

End Sub



2.2 プログラムの機能・構成

      ［ＶＢプロジェクト名］ＢＭＳｏｌｖｅｒＨ

  本ソフトは、多元１次の連立方程式において、係数行列が対称バンドマトリックスとなる

場合にハーフバンドで解くものです。本ソフトの立ち上げ画面は図2-2のメニュー画面です。

同図のように、基本メニューは「データ」、「計算」、「終了」の３つです。「データ」メ

ニューでは入力データを作成し、「計算」メニューでは方程式計算を実行します。そして、

ソフトを終了させる際は、「終了」をクリックします。

                図2-2    「ＢＭＳｏｌｖｅｒＨ」ソフトのメニュー画面

  次に、図2-3に本ソフトのプログラム構成を示します。同図のように、全体を「メニュー」

のSolvHMenuモジュール* で制御し、その下に「データ」および「計算」の２つのプログラ

ム群に分かれています。

    注）*モジュール名と各フォーム名は、ファイル名と名前を一致させてあります。

  データファイルは、「入力データファイル」と計算の「結果データファイル」の２種類を

使用しています。ただし、本図には示していませんが、「計算」部分で一時記憶ファイル

（ファイル拡張子「.tmp」）も使用しています。これらデータファイルはいずれもテキスト

形式であり、エディターや表計算ソフトで閲覧・印刷することができます。

  表2-1に本ソフトのファイルリストを示します。同表で示すように、［データ］の

SolvHDat1～SolvHDat2は、各種データを画面入力するためのものです。特に、SolvHDat2で

は、行列データをグリッド入力できるようにしています。

              図2-3    「ＢＭＳｏｌｖｅｒＨ」ソフトのプログラム構成

               表2-1    「ＢＭＳｏｌｖｅｒＨ」ソフトのファイルリスト

             ［全般］

ファイル名 内  容

SolvHCom.vb 共通変数の宣言とサブルーチンの定義

SolvHMenu.vb メニューの提示･選択

             ［データ］
          

ファイル名 内  容

SolvHDat1.vb データ仕様の入力

SolvHDat2.vb 行列データの入力

AnalDat6.vb データファイルの入出力

             ［計算］
          

ファイル名 内  容

SolvHCalc.vb 使用するデータファイルの設定

SolvHCal3.vb 方程式計算の実行

メニュー

計算

入力データ

ファイル

ファイル

結果データ

データ

1)データ入力

データ

印刷

2)ファイルの入出力

ＡｎａｌＤａｔ６

ＳｏｌｖＨＭｅｎｕ

ＳｏｌｖＨＤａｔ１

ＳｏｌｖＨＤａｔ２

ＳｏｌｖＨＣａｌｃ

ＳｏｌｖＨＣａｌ３



2.3 例題

  次節以降では、次の例題のデータを使用して説明します。

      Ａ=[
2 3 4 0 0 0 0 0
3 5 6 7 0 0 0 0
4 6 9 10 11 0 0 0
0 7 10 12 13 14 0 0
0 0 11 13 15 16 17 0
0 0 0 14 16 18 19 20
0 0 0 0 17 19 21 22
0 0 0 0 0 20 22 23

]                     (2-1)

      ｂ={
10
20
30
40
50
60
70
80

}                                              (2-2)

2.4 「データ」処理プログラム

2.4.1 サブメニュー画面

    「メニュー」プログラムにおける「データ」のサブメニュー選択画面を図2-4に示します。

  同図に示すように、５つのサブメニューがあります。全く新たにデータを作成する場合は

  「データ入力・修正」を選択しますが、既存ファイルを修正して新たなファイルを作成す

  る場合は先に「ファイル読み込み」を行ないます。

    また、「オプション」→「バックアップファイル作成」にチェックを入れると、データ

  読み込み時にバックアップファイル（.bak）が自動作成されます。チェックを外すと作成

  されなくなります。

2.4.2 ファイル読み込み

    サブメニューの「ファイル読み込み」を選択すると、コモンダイアログでファイルを開

  く画面となり、入力データファイルを選定して読み込みます。なお、入力データファイル

  のデフォールトの拡張子は「.dat」です。

                        図2-4    「データ」サブメニュー画面

2.4.3 データ入力・修正

    サブメニューの「データ入力・修正」を選択すると、図2-5のデータ仕様入力画面となり

  ます。本画面の入力について、以下に補足説明します。

    ①タイトルは、任意の題名を入力します。

    ②変数の数とハーフバンド幅を入力します。

        ◇ハーフバンド幅は対角を含む片側バンド幅です（図1-1参照）。

    ③本画面の「設定」ボタンをクリックすると、次の方程式データ入力画面に移ります。

    ④各データ入力が終了すると本画面に戻るので、そこで「すべて確定」ボタンを

      クリックすると入力データが確定します。

                           図2-5    データ仕様入力画面

2.4.4 方程式データ入力画面

    図2-6に方程式データの入力画面を示します。

    ①係数行列の各要素と右辺の値を入力します。

　　      ◇ハーフバンドの部分だけを入力します。*印の部分は入力が無効になります。

    ②「コピー」、「貼り付け」ボタンにより、通常のクリップボード利用ができます。

          ◇「取り消し」ボタンを押すと直前の貼り付けを取り消せます。

    ③すべての行列要素データを入力し終ったなら、「設定」ボタンをクリックして図2-5の

      データ仕様入力画面に戻ります。

                          図2-6    方程式データ入力画面

2.4.5 ファイルへの保存

    サブメニューの「ファイルへの保存」を選択すると、コモンダイアログでファイルを

  保存する画面となり、入力データファイルを設定するとデータが書き出されます。新たな

  ファイル名を入力すると新ファイルが作成されます。先に入力データファイルを読み込ん

  である場合に、同ファイルを選定すると上書きとなります。

    図2-7に本例題の入力データファイルを示します。同図のように、１行目のタイトル以

  降、本例題のデータが書き込まれています。

      ◇１行内の複数データは、「,」（コンマ）で区切られます。

      ◇各行でハーフバンド幅の係数を書き出しているため、最後の方の行では列Nを超える

        部分が0になっています。

                         図2-7    入力データファイル例

2.4.6 最近使用したファイルの選択

    サブメニュー「最近使用したファイル」のポップアップで、ファイルリストから

  選択できます。

  注）新版ソフトでファイルが使用されないとリスト表示されません。

2.5 「計算」プログラム

2.5.1 サブメニュー画面

    「メニュー」プログラムにおける「計算」のサブメニュー画面を図2-8に示します。サブ

  メニューは「計算の実行」と「計算結果の再表示」の２つがあります。「計算の実行」を

  クリックすると、次の「計算」ファイル設定画面に移ります。また、「計算結果の再表

  示」は、計算の実行後にいつでも結果データを再表示するものです。

プログラム名,BMSolverH

ハーフバンド方程式のテストデータ

変数の数, 8

ハーフバンド幅, 3

＜方程式データ＞

行番号、ハーフバンドの各係数、右辺

 1,   2,   3,   4,    10

 2,   5,   6,   7,    20

 3,   9,   10,   11,    30

 4,   12,   13,   14,    40

 5,   15,   16,   17,    50

 6,   18,   19,   20,    60

 7,   21,   22,   0,    70

 8,   23,   0,   0,    80



                        図2-8    「計算」サブメニュー画面

2.5.2 「計算」ファイル設定

    図2-9に「計算」ファイル設定画面を示します。

    ①「入力ファイル指定」ボタンをクリックすると、コモンダイアログで入力データファ

      イルを選定できます。そして、再び「計算」ファイル設定画面に戻ると、「入力ファ

      イル指定」のテキストボックスには選定済み入力ファイル名が、また「出力ファイル

      指定」の方には入力ファイル名の拡張子を「.rep」に変更したファイル名が表示され

      ます。もし、他のファイル名に変更したい場合は、テキストボックスで修正します。

    ②本画面で「設定」ボタンをクリックすると、計算が開始されます。

                          図2-9    使用ファイル設定画面

2.5.3 計算終了

    図2-10に計算終了画面、図2-11に計算確認ダイアログ画面を示します。図2-10では、か

  かった計算時間も表示されますが、本例では0sとなっており、1s未満で終了しています。

  また、図2-11の計算確認ダイアログ画面では、各変数解も表示されます。本画面で

  「ＯＫ」ボタンをクリックすると、立ち上げ時のメニュー画面に戻ります。

      ◇図2-11の各変数解の表示は、最大10変数までで他は省略されます。

                             図2-10    計算終了画面

                          図2-11    計算確認ダイアログ画面

2.5.4 計算結果

    上記2.5.2で指定した出力ファイルに計算結果が書き出されます。図2-12に例題の計算結

  果ファイルへの出力を示します。

                     図2-12    計算結果ファイルへの出力例

＜＜ハーフバンドの方程式計算の結果＞＞

ハーフバンド方程式のテストデータ

［入力データファイル］,C:\VB2022\方程式ｿﾙﾊﾞｰ\BMSolverH\Data\BMSolvHTest.dat

------------------------------------------------------------

＜方程式の解＞

変数番号、   解

 1,   -29.49419

 2,    6.494556

 3,    12.37618

 4,    .2503889

 5,   -.4436424

 6,   -9.032928

 7,    3.634615

 8,    7.856391

------------------------------------------------------------

END    0h 0min 0s



第3章 多元１次方程式

3.1 ガウス消去法の計算

  多元１次方程式の係数行列をフルマトリックスで、ガウス消去法により解きます。ガウス

消去法は、各未知数の列毎に、対角要素のピボットを１に、その他はすべて０としていく

計算方法で、図3-1に手順を略図で示します。

                         図3-1    ガウス消去法の計算方法

  ガウス消去法のプログラムを図3-2に示します。本プログラムについて、○印の番号を付し

た部分について補足説明します。

    ①サブルーチンの引数は、エラー有無kh、変数の数nx、係数行列fmat(,)、右辺兼解

      cvev()です。

        ◇サブルーチンを呼び出す場合は、次のように記述します。

              Call GausCalc(kh, nx, fmat, cvec)

    ②掃出しのピボットが０でないものを探索していますが、元々０にならないことが

      わかっている場合には、この部分は不要です。

    ③ピボット値を設定しています。

    ④ピボットがk行であれば、そのままk行の各要素をピボット値で割っています。

＊＊＊
＊＊＊＊
＊＊＊＊
＊＊＊＊
＊＊＊＊
＊＊＊＊
＊＊＊
＊＊＊
＊＊
＊

＊
＊＊
＊＊ ＊
＊＊ ＊
＊＊ ＊
＊＊ ＊
＊＊ ＊
＊＊ ＊
＊＊ ＊

左辺行列
右辺ベクトル

0)開始行列

＊ ＊＊＊
＊＊＊＊
＊＊＊＊
＊＊＊＊
＊＊＊＊
＊＊＊＊
＊＊＊
＊＊
＊

＊
＊＊
＊＊ ＊
＊＊ ＊
＊＊ ＊
＊＊ ＊
＊＊ ＊
＊＊ ＊
＊＊ ＊

1)第１行を対角要素で割る

＊
＊
＊
＊

＊
＊
＊

＊
＊
＊

＊
＊
＊

＊
＊
＊

＊
＊
＊

１ ＃＃＃ ＃ １

2)第２行以下の第１列を

０にする演算を行なう

＊＊＊ ＊
０
０
０

＃＃＃＃ ＃
＃＃＃＃＃ ＃
＃＃ ＃＃＃＃ ＃

＊＊＊＊
＊＊＊＊
＊＊＊＊
＊＊＊
＊＊
＊

＊＊ ＊
＊＊ ＊
＊＊ ＊
＊＊ ＊
＊＊ ＊
＊＊ ＊

＊
＊
＊

＊

＊
＊

１＊＊＊ ＊

０
０
０

＃＃＃ ＃

3)第２行を対角要素で割る

１
＊＊＊＊＊ ＊
＊＊＊＊＊＊ ＊

１
１

０
０

０
０
０

０
０

０
０

０

００
０

０
０
０

１
１

１
１

１
１

１

０

＊ ＊ ＊＊＊＊ ＊
＊＊＊＊ ＊

＊＊ ＊＊
＊ ＊＊
＊＊
＊
＊

＊
＊＊
＊＊＊
＊＊ ＊ ＊
＊＊ ＊＊＊
＊＊＊ ＊＊＊

＃＃＃＃＃＃
＊ ＊＊＊＊
＊＊＊ ＊
＊＊＊
＊＊
＊

＊
＊＊
＊＊ ＊
＊＊＊＊
＊＊＊＊＊
＊＊＊＊＊ ＊

０
０

０
０
０
０

＊＊＊＊＊＊
＃＃＃＃＃
＃＃＃＃
＃＃＃

＃＃＃＃＃＃ ＃＃＃＃

＃＃＃＃＃＃ ＃＃＃＃

＃＃ ＃＃＃＃ ＃＃＃＃
＃＃ ＃＃＃＃ ＃＃＃＃
＃＃ ＃＃＃＃ ＃＃＃＃
＃＃ ＃＃＃＃ ＃＃＃＃

０
０

０
０
０
０

＃＃＃＃＃
＊＊＊＊＊ ＊

＊＊＊ ＊
＊＊＊
＊＊
＊＊

＊ ＊
＊＊＊
＊＊＊＊
＊＊＊＊＊

１

０
０
０

１＊＊ ＊ ＊
０
０

０

＃＃

4)第２行以外の第２列を

０にする演算を行なう

０
０
０

０
０
０

０
０
０
０
０

０

＃ ＃＃＃＃＃＃

＊＊＊＊＊
＃＃＃ ＃＃＃＃＃＃

＃＃＃ ＃＃＃＃＃＃
＃＃＃ ＃＃＃＃＃＃

＃＃＃ ＃＃＃＃＃＃

＃＃＃ ＃＃＃＃＃＃
＃＃＃ ＃＃＃＃＃＃
＃＃＃ ＃＃＃＃＃＃
＃＃＃ ＃＃＃＃＃＃

０
０

０
０

０
００

００
０ ０
０００
０
０

０

０
０

０

０
０
０

０
００
０００

＃
１

＃
＃
＃
＃
＃
＃
＃
＃

０
００

００ ０
００００

０００００

０ ０００００

０００ ０００００
０００００ ０００

００ ０００００

注）各ステップ毎に、変更された数値を＃で示す。

右辺が解となる

＊

*)最終行・列演算



    ⑤ピボットがk行でなくkk行にあれば、k行とkk行を入れ替えてピボット割算を行ない

      ます。

    ⑥ピボット以外の行を吐き出し計算しています。

    ⑦ピボット行のk列の値をvaluebに設定してます。

    ⑧ピボットの列がピボット以外は０となるようにして、各列を計算しています。

  

                    図3-2（1/2）    ガウス消去法のプログラム

方程式ソルバーK  27

Sub GausCalc(ByRef kh As Integer, ByVal nx As Integer, ByRef fmat(,) As Double, _

ByRef cvec() As Double)                                                       ①

    '線形方程式のガウス消去法計算

    'kh･･･ｴﾗｰ有無（0:なし、-1:有り）

    'nx･･･変数の数

    'fmat(,)･･･係数行列

    'cvec()･･･右辺および解

    Dim aa$

    Dim i, ic, j, k, kk As Integer

    Dim b1, pivotvalue, valueb As Double

    kh = 0

    For k = 1 To nx

        ic = 0  'ﾋﾟﾎﾞｯﾄ有無のflag

        For i = k To nx  '0でないﾋﾟﾎﾞｯﾄを探索                                 ②

            If fmat(i, k) <> 0 Then

                ic = 1

                kk = i

                Exit For

            End If

        Next I

        If ic = 0 Then  'ｴﾗｰ処理

            kh = -1

            aa$ = "pivotvalue=0 AT k=" + Str$(k)

            aa$ = aa$ + Chr(13) + Chr(10) + "計算を終了させます！"

            MsgBox(aa$, 0, "計算エラー")

            Exit Sub

        End If



  

                    図3-2（2/2）    ガウス消去法のプログラム

        pivotvalue = fmat(kk, k)                                              ③

        If k = kk Then  'k行のﾋﾟﾎﾞｯﾄ割算                                      ④

            For j = k To nx

               fmat(k, j) = fmat(k, j) / pivotvalue

           Next j

           cvec(k) = cvec(k) / pivotvalue

        Else    'k行とkk行を入替え、ﾋﾟﾎﾞｯﾄ割算                                ⑤

            For j = k To nx

                b1 = fmat(kk, j)

                fmat(kk, j) = fmat(k, j)

                fmat(k, j) = b1 / pivotvalue

            Next j

            b1 = cvec(kk)

            cvec(kk) = cvec(k)

            cvec(k) = b1 / pivotvalue

        End If

        For i = 1 To nx  'ﾋﾟﾎﾞｯﾄ以外の行を掃出し計算                          ⑥

            If i <> k Then

                valueb = fmat(i, k)                                           ⑦

                If valueb <> 0 Then

                    For j = k To nx

                        fmat(i, j) = fmat(i, j) - valueb * fmat(k, j)         ⑧

                    Next j

                    cvec(i) = cvec(i) - valueb * cvec(k)

                End If

            End If

        Next i

    Next k

End Sub



3.2 プログラムの機能・構成

      ［ＶＢプロジェクト名］ＬｉｎｅＥｑｕａ

  本ソフトは、多元１次の連立方程式をガウス消去法で解くものです。本ソフトの立ち上げ

画面は図3-3のメニュー画面です。同図のように、基本メニューは「データ」、「計算」、

「終了」の３つです。「データ」メニューでは入力データを作成し、「計算」メニューでは

方程式計算を実行します。そして、ソフトを終了させる際は、「終了」をクリックします。

                 図3-3    「ＬｉｎｅＥｑｕａ」ソフトのメニュー画面

  次に、図3-4に本ソフトのプログラム構成を示します。同図のように、全体を「メニュー」

のLineEqMenuモジュール* で制御し、その下に「データ」および「計算」の２つのプログラ

ム群に分かれています。

    注）*モジュール名と各フォーム名は、ファイル名と名前を一致させてあります。

  データファイルは、「入力データファイル」と計算の「結果データファイル」の２種類を

使用しています。ただし、本図には示していませんが、「計算」部分で一時記憶ファイル

（ファイル拡張子「.tmp」）も使用しています。これらデータファイルはいずれもテキスト

形式であり、エディターや表計算ソフトで閲覧・印刷することができます。

  表3-1に本ソフトのファイルリストを示します。同表で示すように、［データ］の

LineEqDat1～LineEqDat2は、各種データを画面入力するためのものです。特に、LineEqDat2

では、行列データをグリッド入力できるようにしています。

            　  図3-4    「ＬｉｎｅＥｑｕａ」ソフトのプログラム構成

                表3-1    「ＬｉｎｅＥｑｕａ」ソフトのファイルリスト

             ［全般］

ファイル名 内  容

LineEquaCom.vb 共通変数の宣言とサブルーチンの定義

LineEqMenu.vb メニューの提示･選択

             ［データ］
          

ファイル名 内  容

LineEqDat1.vb データ仕様の入力

LineEqDat2.vb 行列データの入力

AnalDat6.vb データファイルの入出力

             ［計算］
          

ファイル名 内  容

LineEqCalc.vb 使用するデータファイルの設定

LineEqCal3.vb 方程式計算の実行

入力データ

ファイル

ファイル

結果データ

データ

印刷

計算

データ

1)データ入力

2)ファイルの入出力

ＡｎａｌＤａｔ６

メニュー

ＬｉｎｅＥｑＭｅｎｕ

ＬｉｎｅＥｑＣａｌｃ

ＬｉｎｅＥｑＣａｌ３

ＬｉｎｅＥｑＤａｔ１

ＬｉｎｅＥｑＤａｔ２



3.3 例題

  次節以降では、次の例題のデータを使用して説明します。

      Ａ=[
11 12 20 14 15 16 19 18
21 18 24 26 27 29 30 32
31 34 35 38 39 42 43 46
41 45 46 58 51 55 56 60
51 56 57 62 63 68 69 74
61 67 68 74 75 71 82 88
71 78 79 86 87 94 105 102
81 89 90 98 99 107 108 103

]                 (3-1)

      ｂ={
100
220
330
400
500
600
700
800

}                                             (3-2)

3.4 「データ」処理プログラム

3.4.1 サブメニュー画面

    「メニュー」プログラムにおける「データ」のサブメニュー選択画面を図3-5に示します。

  同図に示すように、５つのサブメニューがあります。全く新たにデータを作成する場合は

  「データ入力・修正」を選択しますが、既存ファイルを修正して新たなファイルを作成す

  る場合は先に「ファイル読み込み」を行ないます。

    また、「オプション」→「バックアップファイル作成」にチェックを入れると、データ

  読み込み時にバックアップファイル（.bak）が自動作成されます。チェックを外すと作成

  されなくなります。

3.4.2 ファイル読み込み

    サブメニューの「ファイル読み込み」を選択すると、コモンダイアログでファイルを開

  く画面となり、入力データファイルを選定して読み込みます。なお、入力データファイル

  のデフォールトの拡張子は「.dat」です。

                        図3-5    「データ」サブメニュー画面

3.4.3 データ入力・修正

    サブメニューの「データ入力・修正」を選択すると、図3-6のデータ仕様入力画面となり

  ます。本画面の入力について、以下に補足説明します。

    ①タイトルは、任意の題名を入力します。

    ②変数の数を入力します。

    ③本画面の「設定」ボタンをクリックすると、次の方程式データ入力画面に移ります。

    ④各データ入力が終了すると本画面に戻るので、そこで「すべて確定」ボタンを

      クリックすると入力データが確定します。

                           図3-6    データ仕様入力画面

3.4.4 方程式データ入力画面

    図3-7に方程式データの入力画面を示します。

    ①係数行列の各要素と右辺の値を入力します。

    ②「コピー」、「貼り付け」ボタンにより、通常のクリップボード利用ができます。

          ◇「取り消し」ボタンを押すと直前の貼り付けを取り消せます。

    ③すべての行列要素データを入力し終ったなら、「設定」ボタンをクリックして図3-6の

      データ仕様入力画面に戻ります。

                          図3-7    方程式データ入力画面

3.4.5 ファイルへの保存

    サブメニューの「ファイルへの保存」を選択すると、コモンダイアログでファイルを

  保存する画面となり、入力データファイルを設定するとデータが書き出されます。新たな

  ファイル名を入力すると新ファイルが作成されます。先に入力データファイルを読み込ん

  である場合に、同ファイルを選定すると上書きとなります。

    図3-8に本例題の入力データファイルを示します。同図のように、１行目のタイトル以

  降、本例題のデータが書き込まれています。

      ◇１行内の複数データは、「,」（コンマ）で区切られます。

                         図3-8    入力データファイル例

3.4.6 最近使用したファイルの選択

    サブメニュー「最近使用したファイル」のポップアップで、ファイルリストから

  選択できます。

  注）新版ソフトでファイルが使用されないとリスト表示されません。

3.5 「計算」プログラム

3.5.1 サブメニュー画面

    「メニュー」プログラムにおける「計算」のサブメニュー画面を図3-9に示します。サブ

  メニューは「計算の実行」と「計算結果の再表示」の２つがあります。「計算の実行」を

  クリックすると、次の「計算」ファイル設定画面に移ります。また、「計算結果の再表

  示」は、計算の実行後にいつでも結果データを再表示するものです。

プログラム名,LineEqua

多元１次方程式のテストデータ

変数の数, 8

＜方程式データ＞

行番号、各係数、右辺

 1,   11,   12,   20,   14,   15,   16,   19,   18,    100

 2,   21,   18,   24,   26,   27,   29,   30,   32,    220

 3,   31,   34,   35,   38,   39,   42,   43,   46,    330

 4,   41,   45,   46,   58,   51,   55,   56,   60,    400

 5,   51,   56,   57,   62,   63,   68,   69,   74,    500

 6,   61,   67,   68,   74,   75,   71,   82,   88,    600

 7,   71,   78,   79,   86,   87,   94,   105,   102,    700

 8,   81,   89,   90,   98,   99,   107,   108,   103,    800



                        図3-9    「計算」サブメニュー画面

3.5.2 「計算」ファイル設定

    図3-10に「計算」ファイル設定画面を示します。

    ①「入力ファイル指定」ボタンをクリックすると、コモンダイアログで入力データファ

      イルを選定できます。そして、再び「計算」ファイル設定画面に戻ると、「入力ファ

      イル指定」のテキストボックスには選定済み入力ファイル名が、また「出力ファイル

      指定」の方には入力ファイル名の拡張子を「.rep」に変更したファイル名が表示され

      ます。もし、他のファイル名に変更したい場合は、テキストボックスで修正します。

    ②本画面で「設定」ボタンをクリックすると、計算が開始されます。

                          図3-10    使用ファイル設定画面

3.5.3 計算終了

    図3-11に計算終了画面、図3-12に計算確認ダイアログ画面を示します。図3-11では、か

  かった計算時間も表示されますが、本例では0sとなっており、1s未満で終了しています。

  また、図3-12の計算確認ダイアログ画面では、各変数解も表示されます。本画面で

  「ＯＫ」ボタンをクリックすると、立ち上げ時のメニュー画面に戻ります。

      ◇図3-12の各変数解の表示は、最大10変数までで他は省略されます。

                             図3-11    計算終了画面

                          図3-12    計算確認ダイアログ画面

3.5.4 計算結果

    上記3.5.2で指定した出力ファイルに計算結果が書き出されます。図3-13に例題の計算結

  果ファイルへの出力を示します。

                     図3-13    計算結果ファイルへの出力例

＜＜多元１次方程式計算の結果＞＞

多元１次方程式のテストデータ

［入力データファイル］,C:\VB2022\方程式ｿﾙﾊﾞｰ\LineEqua\Data\LineEqTest.dat

------------------------------------------------------------

＜方程式の解＞

変数番号、   解

 1,   -35.10623

 2,    5

 3,   -9.428572

 4,   -1.875

 5,    44.68651

 6,   -1.5

 7,    3

 8,   -3.461539

------------------------------------------------------------

END    0h 0min 0s



第4章 １元高次方程式

4.1 Regular Falsi法による解法

  １元高次方程式とは、未知数が１個でその変数xが高次まで含まれる次式で表わされます。

本ソフトでは次式の実数解を算出します。

        f(x)=a0+ a1x+ a2x
2+ ･･･+ anx

n=0                           (4-1)

  １元方程式の解法には種々のものがありますが、ｆ(x)が正と負になる２点xLとxRの間を

どんどん狭めるいわゆる挟み撃ち法、その中でもRegular Falsi法が簡便で優れています。

  図4-1を参照しながら、本解法を説明します。ｆ(xL)とｆ(xR)が異符号となることから、解

がxLとxRの間にあることがわかっているものとします。次に、点Ｐ(xL,ｆ(xL))と点Ｑ

(xR,ｆ(xR))を結ぶ直線を引いて、それがｘ軸と交わる点をxAとします。xAは次式により算出さ

れます。

        xA＝（xL･ｆ(xR)－xR･ｆ(xL)）／（ｆ(xR)－ｆ(xL)）             (4-2)

  そして、ｆ(xA)を計算します。

      ①ｆ(xL)･ｆ(xA)＞０の場合

            解はxAとxRの間にあるから、xAを新たにxLとします。

      ②そうでない場合

            解はxLとxAの間にあるから、xAを新たにxRとします。

  以上の手順を、xAが解に収束するまで、繰り返し計算します。

                         図4-1    Regular Falsi法の概要

  Regular Falsi法によるプログラムを図4-2に示します。本プログラムについて、○印の番

号を付した部分について補足説明します。

    ①ここから⑧までのDoループにおいて、Regular Falsi法の繰返し計算を行ないます。

    ②x下限値のxleftにおける関数値をfleftとします。なお、関数Equavalueは⑨で定義

      しています。

    ③同様に、x上限値のxrightにおける関数値をfrightとします。

    ④点Ｐ(xleft,fleft)と点Ｑ(xright,fright)を結ぶ直線が、ｘ軸と交わる点の座標

      xanswerを計算します。

    ⑤座標xanswerに対応する関数値をfanswerとします。

    ⑥ここから⑦までのIf文で、座標xanswerを次回計算でxleft、xrightのいずれにするか

      を判定します。

    ⑧xanswerの変化が打切り誤差epsより小さくなれば終了させます。

    ⑨１元高次方程式の関数値計算を定義しています。

                  図4-2（1/2）    Regular Falsi法 プログラム

Sub ReguFalsCal(ByRef kh As Integer, ByVal eps As Single, ByRef xanswer As Single, _
ByVal xleft As Single, ByVal xright As Single)
    'Regular Falsi法による計算
    'kh･･･ｴﾗｰ有無（0:なし、-1:有り）
    'eps･･･X解の許容誤差
    'xansb･･･方程式の解
    'xleft･･･変数区間の下限値
    'xright･･･変数区間の上限値
    Dim iter, itermax As Integer
    Dim b1, fanswer, fleft, fright, xansb As Single

    itermax = 100            '繰返し計算の打切り回数
    iter = 0
    kh = -1

    Do                                                                        ①

        fleft = EquaValue(xleft)        'X下限値に対する関数値                ②

        fright = EquaValue(xright)      'X上限値に対する関数値                ③

        xanswer = (xleft * fright - xright * fleft) / (fright - fleft)        ④

        fanswer = EquaValue(xanswer)                                          ⑤

        If fanswer * fright > 0 Then                                          ⑥

             xright = xanswer

        Else

            xleft = xanswer

        End If                                                                ⑦



                   図4-2（2/2）    Regular Falsi法 プログラム

4.2 プログラムの機能・構成

      ［ＶＢプロジェクト名］ＰｏｗｅｒＥｑｕａ

  本ソフトは、１元高次方程式をRegular Falsi法で解くものです。本ソフトの立ち上げ

画面は図4-3のメニュー画面です。同図のように、基本メニューは「データ」、「計算」、

「終了」の３つです。「データ」メニューでは入力データを作成し、「計算」メニューでは

方程式計算を実行します。そして、ソフトを終了させる際は、「終了」をクリックします。

        If iter = itermax Then

            Exit Do

        ElseIf iter > 0 Then

            b1 = Math.Abs(xanswer - xansb)

            If b1 < eps Then                                                 ⑧

                kh = 0

                Exit Do

            End If

        End If

        iter = iter + 1

        xansb = xanswer

    Loop

End Sub

Function EquaValue(ByVal xb As Single) As Single                             ⑨

    '１元高次方程式の関数値

    'xb･･･変数

    Dim i As Integer

    Dim b1 As Single

    b1 = A(0)

    For i = 1 To N

        b1 = b1 + A(i) * xb ^ i

    Next i

    Return b1

End Function



                図4-3    「ＰｏｗｅｒＥｑｕａ」ソフトのメニュー画面

  次に、図4-4に本ソフトのプログラム構成を示します。同図のように、全体を「メニュー」

のPowerEqMenuモジュール* で制御し、その下に「データ」および「計算」の２つのプログラ

ム群に分かれています。

    注）*モジュール名と各フォーム名は、ファイル名と名前を一致させてあります。

  データファイルは、「入力データファイル」と計算の「結果データファイル」の２種類を

使用しています。ただし、本図には示していませんが、「計算」部分で一時記憶ファイル

（ファイル拡張子「.tmp」）も使用しています。これらデータファイルはいずれもテキスト

形式であり、エディターや表計算ソフトで閲覧・印刷することができます。

  表4-1に本ソフトのファイルリストを示します。同表で示すように、［データ］の

PowerEqDat1～PowerEqDat2は、各種データを画面入力するためのものです。特に、

PowerEqDat2では、行列データをグリッド入力できるようにしています。

           　  図4-4    「ＰｏｗｅｒＥｑｕａ」ソフトのプログラム構成

               表4-1    「ＰｏｗｅｒＥｑｕａ」ソフトのファイルリスト

             ［全般］

ファイル名 内  容

PowerEquaCom.vb 共通変数の宣言とサブルーチンの定義

PowerEqMenu.vb メニューの提示･選択

             ［データ］
          

ファイル名 内  容

PowerEqDat1.vb データ仕様の入力

PowerEqDat2.vb 係数データの入力

AnalDat6.vb データファイルの入出力

             ［計算］
          

ファイル名 内  容

PowerEqCalc.vb 使用するデータファイルの設定

PowerEqCal2.vb 計算条件の設定

PowerEqCal3.vb 方程式計算の実行

入力データ

ファイル

ファイル

結果データ

データ

印刷

計算

データ

1)データ入力

2)ファイルの入出力

ＡｎａｌＤａｔ６

メニュー

ＰｏｗｅｒＥｑＭｅｎｕ

ＰｏｗｅｒＥｑＣａｌｃ

ＰｏｗｅｒＥｑＣａｌ２

ＰｏｗｅｒＥｑＤａｔ１

ＰｏｗｅｒＥｑＤａｔ２

ＰｏｗｅｒＥｑＣａｌ３



4.3 例題

  次節以降では、次の例題のデータを使用して説明します。

        f(x)=−5+ 10x−3x
2+ 0.1x

3−0.01x
4+ 0.001x

5=0           (4-3)

4.4 「データ」処理プログラム

4.4.1 サブメニュー画面

    「メニュー」プログラムにおける「データ」のサブメニュー選択画面を図4-5に示します。

  同図に示すように、５つのサブメニューがあります。全く新たにデータを作成する場合は

  「データ入力・修正」を選択しますが、既存ファイルを修正して新たなファイルを作成す

  る場合は先に「ファイル読み込み」を行ないます。

    また、「オプション」→「バックアップファイル作成」にチェックを入れると、データ

  読み込み時にバックアップファイル（.bak）が自動作成されます。チェックを外すと作成

  されなくなります。

                        図4-5    「データ」サブメニュー画面

4.4.2 ファイル読み込み

    サブメニューの「ファイル読み込み」を選択すると、コモンダイアログでファイルを開

  く画面となり、入力データファイルを選定して読み込みます。なお、入力データファイル

  のデフォールトの拡張子は「.dat」です。

4.4.3 データ入力・修正

    サブメニューの「データ入力・修正」を選択すると、図4-6のデータ仕様入力画面となり

  ます。本画面の入力について、以下に補足説明します。

    ①タイトルは、任意の題名を入力します。

    ②方程式の次数を入力します。

    ③本画面の「設定」ボタンをクリックすると、次の方程式データ入力画面に移ります。

    ④各データ入力が終了すると本画面に戻るので、そこで「すべて確定」ボタンを

      クリックすると入力データが確定します。

                           図4-6    データ仕様入力画面

4.4.4 方程式データ入力画面

    図4-7に方程式データの入力画面を示します。

    ①各係数値を入力します。

    ②「コピー」、「貼り付け」ボタンにより、通常のクリップボード利用ができます。

          ◇「取り消し」ボタンを押すと直前の貼り付けを取り消せます。

    ③すべての行列要素データを入力し終ったなら、「設定」ボタンをクリックして図4-6の

      データ仕様入力画面に戻ります。

4.4.5 ファイルへの保存

    サブメニューの「ファイルへの保存」を選択すると、コモンダイアログでファイルを

  保存する画面となり、入力データファイルを設定するとデータが書き出されます。新たな

  ファイル名を入力すると新ファイルが作成されます。先に入力データファイルを読み込ん

  である場合に、同ファイルを選定すると上書きとなります。

    図4-8に本例題の入力データファイルを示します。同図のように、１行目のタイトル以

  降、本例題のデータが書き込まれています。

      ◇１行内の複数データは、「,」（コンマ）で区切られます。

                           図4-7    方程式データ入力画面

                           図4-8    入力データファイル例

4.4.6 最近使用したファイルの選択

    サブメニュー「最近使用したファイル」のポップアップで、ファイルリストから

  選択できます。

  注）新版ソフトでファイルが使用されないとリスト表示されません。

プログラム名,PowerEqua

１元高次方程式のテストデータ

方程式の次数, 5

＜方程式データ＞

係数名、係数値

a0,   -5

a1,    10

a2,   -3

a3,    .1

a4,   -.01

a5,    .001



4.5 「計算」プログラム

4.5.1 サブメニュー画面

    「メニュー」プログラムにおける「計算」のサブメニュー画面を図4-9に示します。サブ

  メニューは「計算の実行」と「計算結果の再表示」の２つがあります。「計算の実行」を

  クリックすると、次の「計算」ファイル設定画面に移ります。また、「計算結果の再表

  示」は、計算の実行後にいつでも結果データを再表示するものです。

                        図4-9    「計算」サブメニュー画面

4.5.2 「計算」ファイル設定

    図4-10に「計算」ファイル設定画面を示します。

    ①「入力ファイル指定」ボタンをクリックすると、コモンダイアログで入力データファ

      イルを選定できます。そして、再び「計算」ファイル設定画面に戻ると、「入力ファ

      イル指定」のテキストボックスには選定済み入力ファイル名が、また「出力ファイル

      指定」の方には入力ファイル名の拡張子を「.rep」に変更したファイル名が表示され

      ます。もし、他のファイル名に変更したい場合は、テキストボックスで修正します。

    ②本画面で「設定」ボタンをクリックすると、次の計算条件入力画面に移ります。

4.5.3 計算条件の入力画面

    図4-11に計算条件入力画面を示します。

    ①X範囲の下限値、上限値、およびX探索の増分を入力します。

        ◇X範囲の下限値からX探索の増分ずつXを増加させることを繰り返して計算します。

        ◇X探索の増分範囲では解が１つしか計算されません。

    ②さらにX解の許容誤差を入力します。

        ◇許容誤差を小さくしすぎると、収束せずに解が算出されないことがあります。

    ③本画面で「設定」ボタンをクリックすると、計算が開始されます。

                          図4-10    使用ファイル設定画面

                             図4-11    計算条件入力画面

4.5.4 計算終了

    図4-12に計算終了画面、図4-13に計算確認ダイアログ画面を示します。図4-12では、か

  かった計算時間も表示されますが、本例では0sとなっており、1s未満で終了しています。

  また、図4-13の計算確認ダイアログ画面では、複数の実数解も表示されます。本画面で

  「ＯＫ」ボタンをクリックすると、立ち上げ時のメニュー画面に戻ります。

      ◇図4-13の複数の実数解の表示は、最大10個で後は省略されます。なお、解の最大数

        はXの次数と同じですが、虚数解は算出されません。

                             図4-12    計算終了画面

                          図4-13    計算確認ダイアログ画面

4.5.5 計算結果

    上記4.5.2で指定した出力ファイルに計算結果が書き出されます。図4-14に例題の計算結

  果ファイルへの出力を示します。

                     図4-14    計算結果ファイルへの出力例

＜＜１元高次方程式計算の結果＞＞

１元高次方程式のテストデータ

［入力データファイル］,C:\VB2022\方程式ｿﾙﾊﾞｰ\PowerEqua\Data\PowerEqTest.dat

X範囲の下限値,-1000

X範囲の上限値, 1000

X探索の増分, .5

------------------------------------------------------------

＜方程式の解＞

No.、   解

 1,    .6092111

 2,    3.022054

 3,    14.34131

------------------------------------------------------------

END    0h 0min 0s



第5章 常微分方程式

5.1 Runge Kutta Gill法による解法

  ここでは下記の１階の常微分方程式を扱います。Runge Kutta Gill法は、本方程式を精度

良く解くための優れた解法です。

        
dy
dx

=f(x,y)                                              (5-1)

         初期条件：x=xoのとき、y=y0                                (5-2)

  xの値をh刻みにとって、次式のxnでのyの値ynを順次算出します。

        xn＝x0＋nh    （n＝1,2,3,・・・）                          (5-3)

  Runge Kutta Gill法では、xn、yn からxn+1、yn+1 を次の公式によって算出します。

        k1=hf(xn,yn) 、r1=(k1−2q0)/2

        η1=yn+r1 、q1=q0+3r1−k1 /2

        k2=hf(xn+h/2,η1) 、r2=(1−√1/2)(k2−q1)

        η2=η1+r2 、q2=q1+3r2−(1−√1/2)k2
        k3=hf(xn+h/2,η2) 、r3=(1+√1/2)(k3−q2)

        η3=η2+r3 、q3=q2+3r3−(1+√1/2)k3
        k4=hf(xn+h,η3) 、r4=(k4−2q3)/6

        η4=η3+r4 、q4=q3+3r4−k4/2                         以上(5-4)

  そして、xn+1＝xn＋h、yn+1＝η4 とします。なお、q0 は出発点x0 では0と置き、あとは１つ前

のステップでのq4 を新しい q0 として(5-4)式を始めます。

  Runge Kutta Gill法によるプログラムを図5-1に示します。本プログラムについて、○印の

番号を付した部分について補足説明します。

    ①c1に √1/2 を入れています。

    ②～③で、刻みhのRunge Kutta Gill法を計算しています。

        ◇EquaValue(xb,yb)は、常微分のf(x,y)に相当し、関数サブルーチンとして別途定

          義しています。

        ◇本部分で変数qとrk、rr、etaは、(5-4)式のqj（j=0～4）とkj、rj、ηj (j=j～4）

          に相当します。

    ④算出されたetaをyb（yn+1 相当）に入れています。

        ◇xn+1 はxb＋hで求まりますが、累積誤差を避けるため、本サブルーチンの呼び出し

          側で別途設定しています。

                     図5-1    Runge Kutta Gill法 プログラム

5.2 プログラムの機能・構成

      ［ＶＢプロジェクト名］ＯｒｄＤｉｆｆＥｑｕａ

  本ソフトは、常微分方程式をRunge Kutta Gill法で解くものです。本ソフトの立ち上げ

画面は図5-2のメニュー画面です。同図のように、基本メニューは「データ」、「計算」、

「終了」の３つです。「データ」メニューでは入力データを作成し、「計算」メニューでは

方程式計算を実行します。そして、ソフトを終了させる際は、「終了」をクリックします。

Sub RKGsolver(ByVal h As Single, ByRef q As Single, _
ByVal xb As Single, ByRef yb As Single)
    'Runge Kutta Gill法
    'h･･･刻み
    'q･･･RKG法の変数q
    'xb･･･X値
    'yb･･･Y値
    Dim b1, c1, eta, rk, rr, xx As Single

    c1 = Math.Sqrt(0.5)                                       ①

    b1 = EquaValue(xb, yb)                                    ②
    rk = h * b1
    rr = 0.5 * (rk - 2 * q)
    eta = yb + rr
    q = q + 3 * rr - 0.5 * rk
    xx = xb + 0.5 * h
    b1 = EquaValue(xx, eta)
    rk = h * b1
    rr = (1 - c1) * (rk - q)
    eta = eta + rr
    q = q + 3 * rr - (1 - c1) * rk
    b1 = EquaValue(xx, eta)
    rk = h * b1
    rr = (1 + c1) * (rk - q)
    eta = eta + rr
    q = q + 3 * rr - (1 + c1) * rk
    xx = xb + h
    b1 = EquaValue(xx, eta)
    rk = h * b1
    rr = (rk - 2 * q) / 6
    eta = eta + rr
    q = q + 3 * rr - 0.5 * rk                                 ③

    yb = eta                                                  ④

End Sub



              図5-2    「ＯｒｄＤｉｆｆＥｑｕａ」ソフトのメニュー画面

  次に、図5-3に本ソフトのプログラム構成を示します。同図のように、全体を「メニュー」

のOrdDiffMenuモジュール* で制御し、その下に「データ」および「計算」の２つのプログラ

ム群に分かれています。

    注）*モジュール名と各フォーム名は、ファイル名と名前を一致させてあります。

  データファイルは、「入力データファイル」と計算の「結果データファイル」の２種類を

使用しています。ただし、本図には示していませんが、「計算」部分で一時記憶ファイル

（ファイル拡張子「.tmp」）も使用しています。これらデータファイルはいずれもテキスト

形式であり、エディターや表計算ソフトで閲覧・印刷することができます。

  表5-1に本ソフトのファイルリストを示します。同表で示すように、［データ］の

OrdDiffDat1～OrdDifDat2は、各種データを画面入力するためのものです。特に、

OrdDiffDat2では、行列データをグリッド入力できるようにしています。

＜対応の常微分方程式＞

    本ソフトでは次式の常微分方程式に対応します。

        
dy
dx

=f(x,y)=a0+a1x+a2x
2+・・・+anx

n+b0xy+b1y+b2y
2

          (5-5)

         ただし、a0、a1、・・・an、b0、b1、b2は係数。

                 xの次数nは任意に指定可能。

       　    図5-3    「ＯｒｄＤｉｆｆＥｑｕａ」ソフトのプログラム構成

             表5-1    「ＯｒｄＤｉｆｆＥｑｕａ」ソフトのファイルリスト

             ［全般］

ファイル名 内  容

OrdDiffCom.vb 共通変数の宣言とサブルーチンの定義

OrdDiffMenu.vb メニューの提示･選択

             ［データ］
          

ファイル名 内  容

OrdDiffDat1.vb データ仕様の入力

OrdDiffDat2.vb 係数データの入力

AnalDat6.vb データファイルの入出力

             ［計算］
          

ファイル名 内  容

OrdDiffCalc.vb 使用するデータファイルの設定

OrdDiffCal2.vb 計算条件の設定

OrdDiffCal3.vb 方程式計算の実行

5.3 例題

  次節以降では、次の例題のデータを使用して説明します。

        
dy
dx

=f(x,y)=10−6x+0.3x2−0.04x
3+0.005x4−0.2xy−0.8y+0.5y2

     (5-6)

  注）上記例題は、すべての係数を設定したテスト用です。一部の係数だけを設定した複数

    の例題について、理論解と一致することを確認しています。

5.4 「データ」処理プログラム

5.4.1 サブメニュー画面

    「メニュー」プログラムにおける「データ」のサブメニュー選択画面を図5-4に示します。

  同図に示すように、５つのサブメニューがあります。全く新たにデータを作成する場合は

  「データ入力・修正」を選択しますが、既存ファイルを修正して新たなファイルを作成す

  る場合は先に「ファイル読み込み」を行ないます。

    また、「オプション」→「バックアップファイル作成」にチェックを入れると、データ

  読み込み時にバックアップファイル（.bak）が自動作成されます。チェックを外すと作成

  されなくなります。

                        図5-4    「データ」サブメニュー画面

5.4.2 ファイル読み込み

    サブメニューの「ファイル読み込み」を選択すると、コモンダイアログでファイルを開

  く画面となり、入力データファイルを選定して読み込みます。なお、入力データファイル

  のデフォールトの拡張子は「.dat」です。

5.4.3 データ入力・修正

    サブメニューの「データ入力・修正」を選択すると、図5-5のデータ仕様入力画面となり

  ます。本画面の入力について、以下に補足説明します。

    ①タイトルは、任意の題名を入力します。

    ②方程式の次数を入力します。これは、Xベキ乗の次数です。

    ③本画面の「設定」ボタンをクリックすると、次のXベキ乗の係数入力画面に移ります。

    ④各データ入力が終了すると本画面に戻るので、そこで「すべて確定」ボタンを

      クリックすると入力データが確定します。

                           図5-5    データ仕様入力画面

5.4.4 Xベキ乗の係数入力画面

    図5-6にXベキ乗の係数入力画面を示します。

    ①Xベキ乗の各係数値a0、a1、・・・を入力します。

    ②「コピー」、「貼り付け」ボタンにより、通常のクリップボード利用ができます。

          ◇「取り消し」ボタンを押すと直前の貼り付けを取り消せます。

    ③すべての行列要素データを入力し終ったなら、「設定」ボタンをクリックして次の

      XYとYベキ乗の係数入力画面に移ります。

5.4.5 XYとYベキ乗の係数入力画面

    図5-7にXYとYベキ乗の係数入力画面を示します。

    ①XYとYベキ乗の各係数値b0、b1、b2 を入力します。

    ②「コピー」、「貼り付け」ボタンにより、通常のクリップボード利用ができます。

          ◇「取り消し」ボタンを押すと直前の貼り付けを取り消せます。

    ③すべての行列要素データを入力し終ったなら、「設定」ボタンをクリックして図5-5の

      データ仕様入力画面に戻ります。

                           図5-6    Xベキ乗の係数入力画面

                        図5-7    XYとYベキ乗の係数入力画面

5.4.6 ファイルへの保存

    サブメニューの「ファイルへの保存」を選択すると、コモンダイアログでファイルを

  保存する画面となり、入力データファイルを設定するとデータが書き出されます。新たな

  ファイル名を入力すると新ファイルが作成されます。先に入力データファイルを読み込ん

  である場合に、同ファイルを選定すると上書きとなります。

    図5-8に本例題の入力データファイルを示します。同図のように、１行目のタイトル以

  降、本例題のデータが書き込まれています。

      ◇１行内の複数データは、「,」（コンマ）で区切られます。

                           図5-8    入力データファイル例

5.4.7 最近使用したファイルの選択

    サブメニュー「最近使用したファイル」のポップアップで、ファイルリストから

  選択できます。

  注）新版ソフトでファイルが使用されないとリスト表示されません。

プログラム名,OrdDiffEqua

常微分方程式テストデータ（総合）

常微分方程式の次数, 4

＜常微分方程式データ＞

係数名、係数値

a0,    10

a1,   -6

a2,    .3

a3,   -.04

a4,    .005

b0,   -.2

b1,   -.8

b2,    .5



5.5 「計算」プログラム

5.5.1 サブメニュー画面

    「メニュー」プログラムにおける「計算」のサブメニュー画面を図5-9に示します。サブ

  メニューは「計算の実行」と「計算結果の再表示」の２つがあります。「計算の実行」を

  クリックすると、次の「計算」ファイル設定画面に移ります。また、「計算結果の再表

  示」は、計算の実行後にいつでも結果データを再表示するものです。

                        図5-9    「計算」サブメニュー画面

5.5.2 「計算」ファイル設定

    図5-10に「計算」ファイル設定画面を示します。

    ①「入力ファイル指定」ボタンをクリックすると、コモンダイアログで入力データファ

      イルを選定できます。そして、再び「計算」ファイル設定画面に戻ると、「入力ファ

      イル指定」のテキストボックスには選定済み入力ファイル名が、また「出力ファイル

      指定」の方には入力ファイル名の拡張子を「.rep」に変更したファイル名が表示され

      ます。もし、他のファイル名に変更したい場合は、テキストボックスで修正します。

    ②本画面で「設定」ボタンをクリックすると、次の計算条件入力画面に移ります。

5.5.3 計算条件の入力画面

    図5-11に計算条件入力画面を示します。

    ①X範囲の最小値、最大値、および分割数を入力します。

        ◇X範囲を分割数で割った刻み毎に解を求めます。

    ②さらに、初期値として、X最小値時のY値を入力します。

        ◇常微分方程式と初期値の関係で、解が発散することがあります。これは、問題

          設定が間違っており、条件を変更して計算し直してください。

    ③本画面で「設定」ボタンをクリックすると、計算が開始されます。

                          図5-10    使用ファイル設定画面

                             図5-11    計算条件入力画面

5.5.4 計算終了

    図5-12に計算終了画面、図5-13に計算確認ダイアログ画面を示します。図5-12では、か

  かった計算時間も表示されますが、本例では0sとなっており、1s未満で終了しています。

  また、図5-13の計算確認ダイアログ画面では、刻み10個までの解とY全範囲の最小・最大

  データが表示されます。本画面で「ＯＫ」ボタンをクリックすると、立ち上げ時のメニュ

  ー画面に戻ります。

                             図5-12    計算終了画面

                          図5-13    計算確認ダイアログ画面

5.5.5 計算結果

    上記5.5.2で指定した出力ファイルに計算結果が書き出されます。図5-14に例題の計算結

  果ファイルへの出力を示します。また、結果ファイルのデータを使用して、グラフ作画ソ

  フトで作成したＸＹグラフを図5-15に示します。

                       図5-14    計算結果ファイルへの出力例

＜＜常微分方程式計算の結果＞＞

常微分方程式テストデータ（総合）

［入力データファイル］,C:\VB2022\方程式ｿﾙﾊﾞｰ\OrdDiffEqua\Data\OrdDiffTestA.dat

X範囲の最小値, 1

X範囲の最大値, 5

X範囲の分割数, 40

X最小値時のY値, 1

------------------------------------------------------------

＜常微分方程式の解＞

X、   f(X)、    Y

 1,    3.765001,    1

 1.1,    3.251313,    1.349948

 1.2,    2.820334,    1.653002

 1.3,    2.437639,    1.915615

 1.4,    2.077399,    2.141274

 1.5,    1.718149,    2.331127

 1.6,    1.340367,    2.484287

 1.7,    .9252597,    2.597957

 1.8,    .454671,    2.66749

 1.9,   -.08762997,    2.686499

 2,   -.7128144,    2.647194

・・・・・・・・・・・・・・・・・・・・

・・・・・・・・・・・・・・・・・・・・

 4,   -.9564104,   -3.048352

 4.1,   -.8834087,   -3.1402

 4.2,   -.8240736,   -3.225469

 4.3,   -.7744738,   -3.305321

 4.4,   -.7317854,   -3.380579

 4.5,   -.6940234,   -3.451828

 4.6,   -.6597958,   -3.519488

 4.7,   -.6281377,   -3.58386

 4.8,   -.5983696,   -3.645165

 4.9,   -.5700059,   -3.703567

 5,   -.5427108,   -3.759189

------------------------------------------------------------

＜最小・最大データ＞

YMin＝-3.759189    （X＝ 5）

YMax＝ 2.686499    （X＝ 1.9）

------------------------------------------------------------

END    0h 0min 0s



                          図5-15    ＸＹグラフ例

                                    注）Plots32使用

常微分方程式の解のグラフ

